怠け者のためのソフトウェアモデル検査入門

安部 達也

TODO

組合せ検証について書き足す。

まえがき

本書は、システムの安全性や活性について検査せざるを得ない状況に陥ってしまったときに最小限のことを 学ぶことで乗り切ろうとする、とにかく必要に迫られない限りは新しい概念を導入せずに、最短経路で目的を 達成しようとする人たちに向けたソフトウェアモデル検査の入門書である。

目次

第1章	逐次プログラム	4
1.1	状態遷移系	4
1.2	モデル検査器 SPIN	5
1.3	命題論理	8
1.4	状態遷移系におけるアサーション	9
第2章	並行プログラム	10
2.1	非决定性	10
2.2	パスの検査	11
2.3	線形時相論理	12
第3章	典型的な検査	15
3.1	停止にかかわる検査	15
3.2	進行性検査	17
3.3	公平性	19
第4章	Büchi オートマトン	21
4.1	定義と諸性質・・・・・・・・・・・・・・・・・・・・・・・・・・・・・・・・・・・・	21
4.2	非空問題の決定可能性	23
4.3	線形時相論理式の変換・・・・・・・・・・・・・・・・・・・・・・・・・・・・・・・・・・・・	23
第5章	効率的な探索	26
5.1	抽象解釈	26
5.2	半順序簡約	29
参考文献		32

第1章

逐次プログラム

逐次プログラムを渡されて、「このシステムでほげほげという性質は常に成り立っているか?」と問われた という状況であると仮定する。

1.1 状態遷移系

渡されたプログラムは

```
1 int x;
2
3 int main() {
4     x = x + 1;
5
6     x = x + 1;
7
8     x = x + 1;
9
10     x = x + 1;
11
12     return 0;
13}
```

といったものであった。システムと大袈裟に言っているが、この程度のプログラムで書かれるものであったとする。しかし、まあシステムと言っているので、このプログラムが表している状態遷移系もここで書いておく。 状態遷移系(state transition system)とは $\mathfrak{S}=(S,s_0,\to,V)$ という四つ組であり、

- 1. S は集合である ($s \in S$ を状態と呼ぶ)
- $2. s_0 \in S$ である(初期状態と呼ぶ)
- $3. \rightarrow \subseteq S \times S$ 、つまり、 \rightarrow は S 上の関係である (遷移関係と呼ぶ)
- 4. $V:S \to \mathfrak{P}(AP)$ 、つまり、V は 状態集合 S から原子命題集合 AP の部分集合への関数である(付値と呼ぶ)

を満たすものをいう。

おっと、原子命題集合 AP なるものを定義していなかった。それは成り立っているかを確かめたい性質が何であるかに依るものであるが、ここでは x>0 や x=0 といったものを原子命題と呼ぶことにし、それらを集めたものを原子命題集合と呼ぶことにし、これは既に与えられているものと仮定する。

前述のプログラムが表している状態遷移系は

と図示される。上記の状態遷移系の定義に従って正確に記述すると、

$$S = \{s_0, s_1, s_2, s_3, s_4\}$$

$$\to = \{\langle s_0, s_1 \rangle, \langle s_1, s_2 \rangle, \langle s_2, s_3 \rangle, \langle s_3, s_4 \rangle\}$$

$$V = \{s_0 \mapsto \{x = 0\}, s_1 \mapsto \{x = 1\}, s_2 \mapsto \{x = 2\}, s_3 \mapsto \{x = 3\}, s_4 \mapsto \{x = 4\}\}$$

である。

1.2 **モデル検査器 SPIN**

このシステムにおいて x の値が常に 0 以上であるかを確かめたいとする。となれば、やはりここは

```
1 int x;
3 int main() {
 6 \quad x = x + 1;
 9 x = x + 1;
 10
12 \quad x = x + 1;
 14
 x = x + 1;
 16
17
 return 0;
18
19 }
```

プリントデバッグである。先頭行と $\mathbf{x} = \mathbf{x} + \mathbf{1}$ がある行ごとに x の値を表示させればよい。 $\mathbf{0}$ 、 $\mathbf{1}$ 、 $\mathbf{2}$ 、 $\mathbf{3}$ 、 $\mathbf{4}$ が表示されるはずである。確かに x の値は常に $\mathbf{0}$ 以上である。しかし、x の値が常に $\mathbf{0}$ 以上であるかを目視

するのも面倒な話である。普段代わりに計算してくれている便利な機械がせっかく目の前にあるのだから、この確認も代行してもらいたいものである。

道具は何でもよいのであるが、ここではモデル検査器 SPIN [2] と呼ばれるものとそのモデリング言語 PROMELA*1 を使うことにする。PROMELA は C に似た構文を持つ。実は意味はあまり似ていないのだ が今のところは気にしないことにする。前述の C プログラムを PROMELA で書くと

```
1 int x;
2
3 active proctype proc() {
4    printf("x: \(\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \('\) \(
```

といったものになる。3 行目が異なるだけである。このまま printf を使い続けては PROMELA で書いた 意義が無いので

```
1 int x;
2
3 active proctype proc() {
4   assert(x >= 0);
5
6   x = x + 1;
7   assert(x >= 0);
8
9   x = x + 1;
10   assert(x >= 0);
11
12   x = x + 1;
13   assert(x >= 0);
```

^{*1} https://spinroot.com/spin/Man/promela.html

```
14
15 \quad x = x + 1;
16 assert(x \ge 0);
17 }
 というように printf の代わりに assert を使う。ここで、
1$ spin -a foo.pml; gcc pan.c; ./a.out
 とすると
1 (Spin Version 6.5.2 -- 15 February 2024)
          + Partial Order Reduction
4 Full statespace search for:
         never claim
                                    - (none specified)
          assertion violations
          acceptance cycles
                                   - (not selected)
          invalid end states
10\,\text{State-vector} 20 byte, depth reached 10, errors: 0
         11 states, stored
          0 states, matched
         11 transitions (= stored+matched)
          O atomic steps
14
 という出力を得る。assertion violations が + であり assert(x >= 0) が検査されていて、そのうえで
 errors: 0 であることを確認する。
   念のため、assert(x > 0) に変更した上で改めて検査してみると
1 pan: wrote foo.pml.trail
3 (Spin Version 6.5.2 -- 15 February 2024)
4\,\mathrm{Warning}\colon\,\mathrm{Search}\,\,\mathrm{not}\,\,\mathrm{completed}
         + Partial Order Reduction
7 Full statespace search for:
                                    - (none specified)
         never claim
          assertion violations
                                   - (not selected)
          acceptance
                       cycles
          invalid end states
11
13 State-vector 20 byte, depth reached 0, errors: 1
```

```
14 1 states, stored
15 0 states, matched
16 1 transitions (= stored+matched)
17 0 atomic steps
```

errors: 1 と出力されており、確かになにかおかしいと言っていることが確認できる。どうせ 4 行目でアサーション違反をしているのであろうが pan: wrote foo.pml.trail と言ってくれているので、実際に違反した実行列(反例と呼ぶ)を確認しておこう。反例は

1\$ spin -t -p foo.pml

で確認できて

なるほど、確かに 4 行目で assert((x>0)) に違反していると報告されている。

1.3 命題論理

今のところ x >= 0 といった単一の原子命題の検査しかおこなっていないが、もっと複雑なことの検査もできる。命題論理式の集合を

- ⊤ は論理式
- 原子命題 p は論理式
- φ が論理式のとき $\neg \varphi$ は論理式である
- φ と ψ が論理式のとき $\varphi \wedge \psi$ は論理式である

を満たす最小の集合と定義する。世の中には Backus-Naur 記法 (BNF) と呼ばれる便利な記法があり、これを使うと、

$$\varphi ::= \top \mid p \mid \neg \varphi \mid \varphi \wedge \varphi$$

と短く書けて大変便利である。状態遷移系 $\mathfrak{S}=(S,s_0,\to,V)$ 中の状態 s が命題論理式 φ を満たすことを

$$\begin{split} \mathfrak{S},s &\vDash \top \\ \mathfrak{S},s &\vDash p \Longleftrightarrow p \in V(s) \\ \mathfrak{S},s &\vDash \neg \psi \Longleftrightarrow \mathfrak{S},s \not\vDash \psi \\ \mathfrak{S},s &\vDash \varphi_0 \land \varphi_1 \Longleftrightarrow \mathfrak{S},s \vDash \varphi_0 \text{ in } \mathfrak{S},s \vDash \varphi_1 \end{split}$$

と定義し $\mathfrak{S},s \models \varphi$ と書く。初期状態 s_0 と遷移関係 \to が $\mathfrak{S},s \models \varphi$ であるかどうかにまったくかかわっていないが、この伏線は 2.3 節で回収する。

1.4 状態遷移系におけるアサーション

1.2 節のとおり assert(x >= 0) はプログラム中に書かれるものであった。これは 1.1 節で導入した状態 遷移系においてはどのように現れてくるであろうか。

本章で扱っているプログラムが表す状態遷移系は前述のとおり

である。assert(x >= 0) はこの各状態において成り立っている原子命題から $x \ge 0$ がいえるかを検査せよと指示しているといえる。実は、このプログラム中の assert(x >= 0) と状態遷移系中の状態におけるアサーション検査が一対一に対応するのは逐次プログラムだけに見られるものであり、並行プログラムでは成立しない。

第2章


並行プログラム

本章では並行性のあるシステムの検査をおこなうために並行プログラムを扱う。

2.1 非決定性

1.1 節で定義した状態遷移系における遷移関係 \to は S 上の関係であって関数であるとは限らず、 $\langle s_0,s_1\rangle,\langle s_0,s_2\rangle\in \to$ (ただし $s_1\neq s_2$) であり得る。こういった非決定性を持つ状態遷移系を生むプログラムの例として

を考える。プロセス proc0 と proc1 はお互いに待合せをしない、つまり、非同期に実行される。これの状態 遷移系を図示すると



というように分岐の存在を確認できる。よって、逐次プログラムを検査するときと異なり、並行プログラムに

おいてはステイトメントの実行列(パスと呼ぶ)は複数存在し得て、この場合、二つのパスが存在する。

プログラム中に assert(x==y) は二つだけ存在するが、この状態遷移系においてこの assert(x==y) が確認されるかもしれない状態は s_1 、 s_2 、 s_3 というように三つある。状態遷移系のある状態において φ が成り立っているかを検査せよという指示が assert(φ) であるが、その状態をプログラム中で指定することは必ずしも容易でない。1.2 節でプリントデバッグに言及したが、逐次プログラムに有効なプリントデバッグが並行プログラムに対しては必ずしも有効でない原因の一つがこれである。

2.2 パスの検査

これまでは「この(特定の)状態において φ という性質が成り立っているか」ということだけを考えてきたが、「このシステムにおいてずっと φ が成り立っているか」ということを考えたいとする。2.1 節の言葉でいうと、「任意のパスにおいて φ がずっと成り立っているか」である。これにはシステムを監視するプロセスを別に立ち上げるとよい。以下のプログラムは

```
1 #define N 5
2 int x;
4 active proctype increment()
5 {
6 do
7 :: atomic { x < N -> x++ }
    od
9 }
10
11 active proctype decrement()
12 {
13
   dο
   :: atomic { x > 0 -> x-- }
16 }
17
18 active proctype monitor() {
    do
   :: assert(0 <= x && x <= N);
20
    od
21
22 }
```

x が N 未満なら x をインクリメントするプロセス increment と x が 0 より大きければ x をデクリメントするプロセス decrement に、ずっと $0 \le x \le N$ であるかを監視し続けるプロセス monitor を追加したものである。PROMELA の構文で初出のものがあるがどういうものかは予想がつくと思うので説明を省略する。このように任意のタイミングで monitor が assert(0 <= x && x <= x N) を実行できるため、SPIN に網

羅的探索してもらうことで「ずっと $0 \le x \le N$ が成り立っているか」を検査できる。これで話が済めば楽なのであるがちょっと込み入った性質を調べたくなると困ったことになる。例えば、「x < 0 である間はずっとy > 0 であるか」といったような性質を調べたいときには、当然 if や do といった制御構文を用いた監視プロセスをつくることはできるものの、調べたい性質から監視プロセスへの翻訳という手間がかかる。これの原因は、調べたい性質が宣言的に与えられているにもかかわらず、監視プロセスは手続き的に書かれなければならないからである。この手間を失くすために、モデル検査器 SPIN に調べたい性質を直接与えることにする。

2.3 線形時相論理

1.3 節で定義した命題論理式を

$$\varphi \coloneqq \top \mid p \mid \neg \varphi \mid \varphi \wedge \varphi \mid \mathbf{X} \varphi \mid \varphi \to \mathbf{U} \varphi$$

と拡張したものを線形時相論理(Linear Temporal Logic, LTL)式という。任意の i で $\langle s_i, s_{i+1} \rangle \in \to$ である状態の無限列 $\pi = s_0 s_1 \cdots$ をパスと呼ぶ。有限列は最後の状態を繰り返し続けること(stuttering という)で無限列と見なせる。3.1 章で停止性に関する検査をする場合は stuttering を都合よく無視したりする。 π 中の i 番目の状態を $\pi(i)$ と書く。 π の i 番目以降を $\pi \upharpoonright i$ と書く。 $(\pi \upharpoonright i)(0)$ は s_i である。状態遷移系 $\mathfrak{S} = (S, \to, V)$ 中のパス π が線形時間論理式 φ を満たすことを

 $\mathfrak{S}, \pi \models \top$

 $\mathfrak{S}, \pi \vDash p \iff p \in V(\pi(0))$

 $\mathfrak{S}, \pi \vDash \neg \psi \iff \mathfrak{S}, \pi \nvDash \psi$

 $\mathfrak{S}, \pi \vDash \varphi_0 \land \varphi_1 \iff \mathfrak{S}, \pi \vDash \varphi_0$ かつ $\mathfrak{S}, \pi \vDash \varphi_1$

 $\mathfrak{S}, \pi \models X \psi \iff \mathfrak{S}, \pi \upharpoonright 1 \models \psi$

 $\mathfrak{S}, \pi \models \varphi_0 \cup \varphi_1 \iff$ ある i が存在して $\mathfrak{S}, \pi \restriction i \models \varphi_1$ かつ任意の j < i で $\mathfrak{S}, \pi \restriction j \models \varphi_0$

で定義する。 $\Diamond \psi$ と $\Box \psi$ を $\top U \psi$ と $\neg \Diamond \neg \psi$ にそれぞれ定義することで

 $\mathfrak{S}, \pi \models \Diamond \psi \iff$ ある i が存在して $\mathfrak{S}, \pi \restriction i \models \psi$

 $\mathfrak{S}, \pi \vDash \Box \psi \iff$ 任意の i で $\mathfrak{S}, \pi \upharpoonright i \vDash \psi$

である。1.3 節では特定の状態が論理式を満たすかどうかといったことしか考えることができず、以下のように、パスを満たすかどうかといったことを考えることができなかった。

命題 2.1. φ を命題論理式とする。このとき、任意の π,π' に対して $\pi(0)=\pi'(0)$ ならば $\mathfrak{S},\pi \models \varphi$ と $\mathfrak{S},\pi' \models \varphi$ は同値である。

証明. $\mathfrak{S}, \pi \models \varphi$ として一般性を失わない。 φ の構成に関する帰納法で示す。 φ の形で場合分けをする。 φ が \top のとき、ただちに従う。

 φ が p のとき、 $\pi(0) = \pi'(0)$ よりただちに従う。

 φ が ¬ ψ のとき、 $\mathfrak{S}, \pi \models \neg \psi$ より $\mathfrak{S}, \pi \not\models \neg \psi$ である。帰納法の仮定より $\mathfrak{S}, \pi' \not\models \neg \psi$ である。よって、 $\mathfrak{S}, \pi \models \neg \psi$ である。

 φ が $\varphi_0 \land \varphi_1$ のとき、 $\mathfrak{S}, \pi \models \varphi_0 \land \varphi_1$ より $\mathfrak{S}, \pi \models \varphi_0$ かつ $\mathfrak{S}, \pi \models \varphi_1$ である。帰納法の仮定より $\mathfrak{S}, \pi' \models \varphi_0$

かつ $\mathfrak{S}, \pi' \models \varphi_1$ である。よって、 $\mathfrak{S}, \pi' \models \varphi_0 \land \varphi_1$ である。

その一方、線形時相論理式に対してはパスがそれを満たすかを考えることができるようになっている。 監視するプロセスの代わりに

```
1#define N 5
2 int x;
4 active proctype increment()
5 {
   do
7 :: atomic { x < N \rightarrow x++ }
   od
9 }
11 active proctype decrement()
12 {
13
   do
14 :: atomic { x > 0 -> x-- }
   od
15
16 }
17
18 ltl { [](0 <= x && x <= N) }
```

というように SPIN に線形時相論理式を与えることでも検査できる。1t1 $\{\varphi\}$ の検査をするにあたっては 4.2 節で紹介する acceptance cycle checking を足さないといけないため、網羅的探索のコマンドに

```
1$ spin -a foo.pml; gcc pan.c; ./a.out -a
```

というように -a というオプションをつける必要がある。

線形時相論理式により安全性や活性と呼ばれる性質を記述できる。安全であるとは、危険な状況をあらかじめ決めておいた場合にその状況に至らない(到達可能でない)ことをいう。危険な状況を φ であるとすると、これに至らないとは $\neg \diamondsuit \varphi$ である。もちろんこれと同値である、ずっと安全であるを意味する $\Box \neg \varphi$ でもよい。反対に活性とは、こうなってほしいという状況をあらかじめ決めておいた場合にその状況にいずれ至ることをいう。こうなってほしいという状況を φ であるとすると、これにいずれ至るとは $\diamondsuit \varphi$ である。線形時相論理式 $\varphi \ U \ \psi$ を用いることで、より込み入った性質も検査できるようになった。

ちなみに、線形時相論理式の表現力にもある種の限界がある。 $V(\pi)$ をパス π 中の状態 s_i に V をそれぞれ適用して得られる原子命題集合の列とする(トレースと呼ぶ)。任意のパス π で $\mathfrak{S}, \pi \models \varphi$ であることを $\mathfrak{S}, s_0 \models \varphi$ と書くことにする。以下のように、線形時相論理式はトレースの集合が一致している状態遷移系を 区別できない。

命題 2.2. φ を線形時相論理式とする。 $\mathfrak{S}_0,\mathfrak{S}_1$ のトレース集合が一致しているとき $\mathfrak{S}_0 \models \varphi$ と $\mathfrak{S}_1 \models \varphi$ は同値である。

証明. \mathfrak{S}_0 と \mathfrak{S}_1 の付値をそれぞれ V_0 と V_1 と書くことにする。

 $\mathfrak{S}_0 \models \varphi$ として一般性を失わない。 π_1 を \mathfrak{S}_1 上にとる。 $\mathfrak{S}_1 \models \varphi$ の真偽は $V_1(\pi_1)$ で決まる。

仮定より、 $V_1(\pi_1)$ と一致する π_0 が \mathfrak{S}_0 上に存在する。 $\mathfrak{S}_0 \models \varphi$ より $\mathfrak{S}_0, \pi_0 \models \varphi$ である。 $V_0(\pi_0)$ と $V_1(\pi_1)$ は一致しているので $\mathfrak{S}_1, \pi_1 \models \varphi$ である。 π_1 は任意にとったので $\mathfrak{S}_1 \models \varphi$ である。

これらを区別したければそういった論理も存在するし、また、それと線形時相論理の両方を包含する、より 表現力の強い論理も存在する。

第3章

典型的な検査

検査したい性質をユーザが与えるものだけでなく、知りたい性質としてよく挙げられるものの検査について 紹介する。

3.1 停止にかかわる検査

SPIN はプロセスが遷移ができなくなるとそれを異常と判断して報告する。止まってしまってもよい (SPIN に見逃してほしい) 箇所には end ラベルをつける。以下は最大公約数を求めるプログラムであり

4 行目で x = y であって条件を満たすことができないことを利用してプログラムを停止させている。

これまでに停止するプログラムの例を挙げてきたが、SPIN がエラーを出力していないのは SPIN がプログラム中に end ラベルが無いときはプロセスの最終行に end ラベルがあるものとすると仮定しているからである。

以下はいわゆるデッドロックするプログラム

```
1 mtype = {LOCKED, UNLOCKED};
2 mtype mutex0 = UNLOCKED;
3 mtype mutex1 = UNLOCKED;
4
5 inline lock(m){
6  atomic { m == UNLOCKED -> m = LOCKED }
7 }
```

```
9 inline unlock(m){
  m = UNLOCKED
11 }
12
13 active proctype proc0()
14 {
15 repeat:
   lock(mutex0);
   lock(mutex1);
18
   unlock(mutex0);
19
   unlock(mutex1);
   goto repeat
22 }
24 active proctype proc1()
25 {
26 repeat:
   lock(mutex1);
   lock(mutex0);
29
   unlock(mutex1);
   unlock(mutex0);
   goto repeat
32
33 }
 である。実際検査すると pan:1: invalid end state (at depth 1) という出力が得られる。その一方、
1 active proctype proc1()
2 {
3 repeat:
   lock(mutex0);
   lock(mutex1);
   unlock(mutex0);
  unlock(mutex1);
   goto repeat
10 }
```

というように proc1 のロックをとる順番を変更すると invalid な end 状態に至って止まってしまった、ということは起きなくなる。

3.2 進行性検査

起こってほしいことが本当に起こる(進行性)の検査を紹介する。起こってほしいことの行に progress ラベルをふる。このラベルをふった行を通らないパスが存在するとき SPIN は異常と判断する。

```
1 mtype = {LOCKED, UNLOCKED};
2 mtype mutex0 = UNLOCKED;
3 mtype mutex1 = UNLOCKED;
5 inline lock(m){
6 atomic { m == UNLOCKED -> m = LOCKED }
7 }
9 inline unlock(m){
10 m = UNLOCKED
11 }
13 active proctype proc0()
14 {
15 repeat:
  lock(mutex0);
17 lock(mutex1);
18
19 progress:
  skip;
21
22 unlock(mutex0);
23 unlock(mutex1);
24 goto repeat
25 }
27 active proctype proc1()
28 {
29 repeat:
30 lock(mutex0);
  lock(mutex1);
33 unlock(mutex1);
34 unlock(mutex0);
```

```
35 goto repeat
36 }
```

といったプログラムを

```
1 \$ spin -a foo.pml; gcc -DNP pan.c; ./a.out -l
```

で検査する。進行性の検査には (non-progress を意味する NP をオンにする) -DNP と (liveness を意味する) -1 オプションが必要である。

```
1 pan:1: non-progress cycle (at depth 2)
```

という出力を得る。これは non-progress、つまり progress ラベルを通らない、cycle、つまりある状態からその状態へのサイクルが存在することを意味し、すなわち、いくら待っても progress ラベルを通ることがないパスが存在するということである。

具体的には proc1 が延々とロックをとり続けていて proc0 に譲っていない。そこで、手番(turn)を考慮する。つまり、

```
1 \text{ bit turn0} = 0;
2 bit turn1 = 0;
4 inline lock(m,t){
   atomic { m == UNLOCKED && t == _pid -> m = LOCKED }
6 }
8 inline unlock(m,t){
9 m = UNLOCKED; t = 1 - _pid
10 }
11
12 active proctype proc0()
14 repeat:
   lock(mutex0, turn0);
    lock(mutex1, turn1);
17
18 progress:
    skip;
19
    unlock(mutex0, turn0);
21
   unlock(mutex1, turn1);
    goto repeat
23
24 }
25
```

```
26 active proctype proc1()
27 {
28 repeat:
29  lock(mutex0, turn0);
30  lock(mutex1, turn1);
31
32  unlock(mutex0, turn0);
33  unlock(mutex1, turn1);
34  goto repeat
35 }
```

とすれば、一度ロックをとったら次は譲るので進行性が保証される。

3.3 公平性

前節で手番を導入することで譲るを実現したが、そういったある種の公平性を与えたうえで検査するという こともできる。例として

```
1 active proctype proc0()
2 {
3 do
4 :: 1;
5 chosen:
      printf("0-0\n");
7 progress:
      printf("0-1\n");
  od;
10 }
11
12 active proctype proc1()
13 {
   do
14
   :: printf("1-0\n");
15
       printf("1-1\n");
   od;
17
18 }
```

を考える。PROMELA の実行モデルがインターリービングであり、proc1 が隙間なく実行し続けていると proc0 が実行する間隙を与えてもらえず progress ラベルがふってある行が実行されない。

SPIN に -f というオプションがあり、これをつけると、ずっと実行可能であればいずれ実行される、という仮定(弱公平性という)を加えられる。

1\$ spin -a foo.pml; gcc -DNP pan.c; ./a.out -l -f

この仮定の下で進行性が保証される。

弱公平性の仮定だけは SPIN のオプションで置いたうえで検査できるが、それ以外の公平性については線 形時相論理式を与えて検査をおこなう。以下のプログラム

```
1 active proctype proc0()
2 {
   dο
4 :: lock(mutex);
5 chosen:
       printf("0-0\n");
7 here:
       printf("0-1\n");
       unlock(mutex);
   od;
10
11 }
12
13 active proctype proc1()
14 {
   dο
   :: lock(mutex);
16
       printf("1\n");
17
       unlock(mutex);
18
   od;
19
20 }
```

を考える。このとき □◇proc0@here、つまり、here が無限回実行される、は成り立たない。というのも、proc1 が実行し続けるパスが存在するからである。

無条件公平性 □◇ proc0@chosen、つまり、chosen が無限回実行される、という仮定を置く。このとき □◇ proc0@here は成り立つ。というのも、chosen が実行されるその直後に here も実行されるからである。 弱公平性 ¬◇□(mutex == UNLOCKED ∧ ¬proc0@chosen)、つまり、chosen がずっと実行可能であるならば chosen がいずれは実行される、という仮定を置く。このとき □◇ proc0@here は成り立たない。というのも、ずっと mutex == UNLOCKED はたびたび実行不可能になるからである。

その一方、強公平性 \neg (\square \diamondsuit mutex == UNLOCKED $\land \diamondsuit$ \square \neg proc0@chosen)、つまり、chosen が無限回実行可能であるならば chosen がいずれは実行される、という仮定を置く。このとき \square \diamondsuit proc0@here が成り立つ。というのも、というのも、その無限回 chosen が実行されるその直後に here も実行されるからである。

第4章

Büchi オートマトン

状態遷移系において線形時相論理式を満たすかどうかがどのように検査されているかの理論を紹介する。

4.1 定義と諸性質

有限オートマトン $\mathfrak{A} = (\Sigma, S, I, \delta, F)$ とは、

- 1. Σ は有限集合 ($a \in \Sigma$ をアルファベットと呼ぶ)
- 2. S は有限集合 ($s \in S$ を状態と呼ぶ)
- 3. $I \subseteq S$ は初期状態集合
- $4. \ \delta \$ は $S \times \Sigma \$ から $\mathfrak{P}(S)$ への関数
- $5. F \subseteq S$ は終了状態集合

をいう。アルファベットの有限列全体の集合を Σ^* と書く。 $w=a_0a_1\cdots a_{n-1}\in \Sigma^*$ が $\mathfrak A$ に受理されるとはある s_0,s_1,\ldots,s_n が存在して

- $s_0 \in I$
- 任意の $1 \le i < n$ について $s_i \in \delta(s_{i-1}, a_{i-1})$
- $s_{n-1} \in F$

をいう。 $\mathfrak A$ が受理されるもの全体を $L(\mathfrak A)$ と書く。

Büchi オートマトンとは有限オートマトンをアルファベットの無限列を受理できるように拡張したものをいう。アルファベットの可算無限列全体の集合を Σ^ω と書く。 $w=a_0a_1\dots\in \Sigma^\omega$ が $\mathfrak A$ に受理されるとはある s_0,s_1,\dots が存在して

- $s_0 \in I$
- 任意の $1 \le i$ について $s_i \in \delta(s_{i-1}, a_{i-1})$
- $\{s \mid \#(\{i \mid s_i = s\}) = \omega\} \cap F \neq \emptyset$


をいう。ただし、最後の条件中の#は集合の濃度をとる関数であり、 ω は最小の極大順序数である。この三つ目の条件は終了状態が無限回現れるということを表している。

例 4.1. $\mathfrak{A} = (\{a,b\},\{s_0,s_1\},\{s_0\},\{\langle s_0,a\rangle\mapsto\{s_1\},\langle s_0,b\rangle\mapsto\{s_0\},\langle s_1,a\rangle\mapsto\{s_1\},\langle s_1,b\rangle\mapsto\{s_0\}\},\{s_1\})$

$$a$$
 s_0
 a
 s_1

とすると $L(\mathfrak{A})$ は a を無限回通るもの全体、つまり、 $\{c_0c_1\cdots|\#(\{i\mid c_i=a\})=\omega\}$ である。

例 4.2. $\mathfrak{A} = (\{a,b\},\{s_0,s_1\},\{s_0\},\{\langle s_0,a\rangle\mapsto\{s_0\},\langle s_0,b\rangle\mapsto\{s_0,s_1\},\langle s_1,b\rangle\mapsto\{s_1\}\},\{s_1\})$



とすると $L(\mathfrak{A})$ は a を有限回通るもの全体、つまり、 $\{c_0c_1\cdots \mid \#(\{i\mid c_i=a\})<\omega\}$ である。

二つの Büchi オートマトンについてそれらが受理するものの合併を受理する Büchi オートマトンをつくれることは以下のように容易にわかる。

命題 4.3. Büchi オートマトン $\mathfrak{A}_i=(\Sigma,S_i,I_i,\delta_i,F_i)$ i=0,1 に対して $L(\mathfrak{A})=L(\mathfrak{A}_0)\cup L(\mathfrak{A}_1)$ となる Büchi オートマトン $\mathfrak A$ が存在する。

証明. \mathfrak{A} を $(\Sigma, (S_0 \times \{0\}) \cup (S_1 \times \{1\}), (I_0 \times \{0\}) \cup (I_1 \times \{1\}), \delta, (F_0 \times \{0\}) \cup (F_1 \times \{1\}))$ 、ただし

$$\delta(s, a) = \begin{cases} \delta_0(s_0, a) \times \{0\} & \text{if } s = \langle s_0, 0 \rangle \\ \delta_1(s_1, a) \times \{1\} & \text{if } s = \langle s_1, 1 \rangle \end{cases}$$

にとれば十分である。

その一方、共通部分については一考を要する。まず、有限オートマトンについて以下が成り立つ。

命題 4.4. 有限オートマトン $\mathfrak{A}_i=(\Sigma,S_i,I_i,\delta_i,F_i)$ i=0,1 に対して $L(\mathfrak{A})=L(\mathfrak{A}_0)\cap L(\mathfrak{A}_1)$ となる有限 オートマトン $\mathfrak A$ が存在する。

証明. $\mathfrak A$ を $(\Sigma, S_0 \times S_1, I_0 \times I_1, \delta_0 \times \delta_1, F_0 \times F_1)$ にとれば十分である。

受理するものの共通部分を受理する Büchi オートマトンをつくるにあたっても基本的なアイデアは同じである。しかし、Büchi オートマトンの受理条件は有限オートマトンの受理条件と異なり、安直に終了状態集合の直積をとるとそれによって得られる受理条件は強すぎるものとなる。というのも、無限回通る状態の集合が一致していないといけないという要請になってしまうからである。そこで、この部分に微調整をおこなって以下を得る。

命題 4.5. Büchi オートマトン $\mathfrak{A}_i=(\varSigma,S_i,I_i,\delta_i,F_i)$ i=0,1 に対して $L(\mathfrak{A})=L(\mathfrak{A}_0)\cap L(\mathfrak{A}_1)$ となる Büchi オートマトン $\mathfrak A$ が存在する。

証明. まず、状態集合について $S_0 \times S_1 \times \{0,1\}$ というように、単なる $S_0 \times S_1$ でなく $\{0,1\}$ と直積をとることで $S_0 \times S_1$ のコピーをつくる。これは遷移によって状態のモードを 0 というモードと 1 というモードと に切り替えることがあるということだと考えてほしい。

初期状態集合を $I_0 \times I_1 \times \{0\}$ とする。モード 0 から開始するというつもりである。

遷移関数 $\delta(\langle s_0, s_1, i \rangle, a)$ を $\delta_0(s_0, a) \times \delta_1(s_1, a) \times \{ \text{mode}(s_0, s_1, i) \}$ とする。ただし、

$$\mathrm{mode}(s_0,s_1,i) = \begin{cases} 0 & \text{if } (i=0 \text{ かつ } s_0 \not\in F_0) \text{ または } (i=1 \text{ かつ } s_1 \in F_1) \\ 1 & \text{if } (i=0 \text{ かつ } s_0 \in F_0) \text{ または } (i=1 \text{ かつ } s_1 \not\in F_1) \end{cases}$$

である。モードが 0 であるとして、モードを 1 に切り替えるのは状態が F_0 に入ったときであり、また、そうでなければモードを 0 に維持しておく、ということである。逆に、モードが 1 であった場合も同様である。

終了状態集合を $F_0 \times S_1 \times \{0\}$ とする。 \mathfrak{A}_0 の受理条件を満たすようになっていることはすぐにわかるだろう。その一方、以下のとおり \mathfrak{A}_1 の受理条件も満たしている。 δ の定義により F_0 に入った瞬間にモードは 1 に変わり、 F_1 に入らないとモード 0 に戻ることはなく、よって、 $F_0 \times S_1 \times \{0\}$ を無限回通らなければならないという条件は F_1 も無限回通らなければならないということをも要請している。

4.2 非空問題の決定可能性

Büchi オートマトンは有限状態であるもののその上には無限遷移が存在するので、受理または非受理の判定が決定可能であるかはただちにはわからない。しかし、アルファベット集合の有限性より一つの状態からの分岐は有限であることに注目するとわりと容易に非空問題については決定可能であることがわかる。

定理 4.6. $L(\mathfrak{A}) \neq \emptyset$ かどうかは決定可能である。

証明.まず、受理条件は終了状態を無限回通ることなので一回は通る必要がある。初期状態から終了状態への同じ状態を二度以上通らないという制約の下での到達可能性は前述の有限分岐性より有限回で判定できる。

受理条件は終了状態を無限回通ることなので、次はその終了状態からまた自身に戻ってくることができるかを確認する(acceptance cycle checking)。これもさきほどと同様に有限回で判定できる。

4.3 線形時相論理式の変換

U だけを持つ線形時相論理式 φ に対して、この φ だけを受理する Büchi オートマトンを \mathfrak{A}_{φ} を構築する。まず、受理状態集合 F を F_0,\ldots,F_{k-1} と複数に増やした上で受理条件を、 $w=a_0a_1\cdots\in \Sigma^{\omega}$ が $\mathfrak A$ に対してある s_0,s_1,\ldots が存在して

- $s_0 \in I$
- 任意の $1 \le i$ について $s_i \in \delta(s_{i-1}, a_{i-1})$
- 任意の $0 \le j < k$ に対して $\{s \mid \#(\{i \mid s_i = s\}) = \omega\} \cap F_j \neq \emptyset$

に拡張したものを一般化 Büchi オートマトンという。以下が成り立つ。

定理 4.7. 一般化 Büchi オートマトン $\mathfrak A$ に対して $L(\mathfrak A)=L(\mathfrak A')$ となる Büchi オートマトン $\mathfrak A'$ が存在する。

証明. 命題 4.5 でおこなった Büchi オートマトンの構築を終了状態集合が二つから任意有限個に増えたという設定の下でおこなえば十分である。 □

次に、 $cl(\varphi)$ を以下で定義する。

• $\top \in cl(\varphi)$

- $\varphi \in cl(\varphi)$
- $\psi \in cl(\varphi)$ $\Leftrightarrow \psi \in cl(\varphi)$
- $\neg \psi \in cl(\varphi)$ ならば $\psi \in cl(\varphi)$
- $\varphi_0 \land \varphi_1 \in cl(\varphi)$ ならば $\varphi_0 \in cl(\varphi)$ かつ $\varphi_1 \in cl(\varphi)$
- $X\psi \in cl(\varphi)$ ならば $\psi \in cl(\varphi)$
- $\varphi_0 \cup \varphi_1 \in cl(\varphi)$ ならば $\varphi_0 \in cl(\varphi)$ かつ $\varphi_1 \in cl(\varphi)$

 $s \subseteq cl(\varphi)$ が $cl(\varphi)$ -極大無矛盾であるとは


- $\psi \in cl(\varphi)$ について、 $\psi \in s$ と $\neg \psi \notin s$ が同値
- $\varphi_0 \land \varphi_1 \in cl(\varphi)$ について、 $\varphi_0 \land \varphi_1 \in s$ と $\varphi_0 \in s$ かつ $\varphi_1 \in s$ が同値

をいう。

最後に、 $\mathfrak{A}_{\varphi} = (\mathfrak{P}(AP(\varphi)), \{s \mid s \text{ は } cl(\varphi)\text{-極大無矛盾}\}, \{s \mid \varphi \in s\}, \delta, \{F_{\varphi_0 \cup \varphi_1} \mid \varphi_0 \cup \varphi_1 \in cl(\varphi)\})$ さする。ただし、

- $AP(\varphi)$ は φ に現れる原子命題全体という有限集合である。
- $s' \in \delta(s,a)$ を以下のすべてを満たすものと定義する:
 - $-a = s \cap AP$
 - $\mathbf{X}\psi \in cl(\varphi)$ について、 $\mathbf{X}\psi \in s$ と $\psi \in s'$ が同値
 - $-\varphi_0 \cup \varphi_1 \in cl(\varphi)$ について
 - * $\varphi_0 \cup \varphi_1 \in s$
 - * $\varphi_1 \in s$ または $[\varphi_0 \in s \text{ かつ } \varphi_0 \cup \varphi_1 \in s']$

が同値。以下の図中で二重に囲まれたものからはすべてのものに矢印が出ているものとする。



• $F_{\varphi_0 \cup \varphi_1}$ $\mathcal{E} \{ s \mid \varphi_1 \in s \text{ stat } \neg (\varphi_0 \cup \varphi_1) \in s \} \text{ bts.}$

考え方を説明する。まず、状態の定義に $cl(\varphi)$ -極大無矛盾性を利用することで X や U を持たない論理式たちの整合性は保証されている。その一方、一般の論理式たちとの整合性がとれているとは限らない。そこで、遷移をこのように定義することで X や U を持つ論理式との整合性もとろうとしている。しかし、U を持つ論理式については遷移の定義の工夫だけではまだ足りていない。そこで、終了状態集合族をうまく定義することで全体としてうまくいくようにしている、と考えるとよい。

 φ_0 U φ_1 について、これが成り立つものであるかはいずれ $\varphi_1 \in s$ となるかに依る。このことが $F_{\varphi_0 \cup \varphi_1}$ の 定義中の $\varphi_1 \in s$ に対応している。 $\neg (\varphi_0 \cup \varphi_1)$ については、まず、遷移の定義より以下が成り立つことに注意する。

命題 4.8. $\neg (\varphi_0 \cup \varphi_1) \in cl(\varphi)$ について、

1. $\neg (\varphi_0 \cup \varphi_1) \in s$

2. $\neg \varphi_1 \in s$ かつ $[\neg \varphi_0 \in s$ または $\neg (\varphi_0 \cup \varphi_1) \in s']$

は同値である。

証明. $cl(\varphi)$ -極大無矛盾性より $\varphi_0 \cup \varphi_1 \not\in s$ である。 δ の定義より $\varphi_1 \not\in s$ かつ $[\varphi_0 \not\in s$ または $\varphi_0 \cup \varphi_1 \not\in s']$ である。 $cl(\varphi)$ -極大無矛盾性より $\neg \varphi_1 \in s$ かつ $[\neg \varphi_0 \in s]$ または $\neg (\varphi_0 \cup \varphi_1) \in s']$ である。逆も同様である。

よって、一般化 Büchi オートマトンの受理条件が終了状態集合族を無限回通らなければならないというものであることを考慮すると、 $\neg (\varphi_0 \cup \varphi_1)$ についての終了状態集合の条件は $\neg (\varphi_0 \cup \varphi_1) \in S$ ということになる。

定理 4.9. $\#(S)<\omega$ であり、各 $s\in S$ に対して $\#(\{s'\mid \langle s,s'\rangle\in \to\})<\omega$ である $\mathfrak{S}=(S,s_0,\to,V)$ について $\mathfrak{S}\models\varphi$ かどうかは決定可能である。

証明. $\mathfrak{A}_{\mathfrak{S}}$ を $(\mathfrak{P}(AP(\varphi)), S, \{s_0\}, \delta, S)$ と定義する。ただし、

$$\delta(s,a) = \begin{cases} \{ \, s' \mid \langle s,s' \rangle \in \, \to \, \} & \text{if } V(s) = a \\ \varnothing & \text{otherwise} \end{cases}$$

である。

構成の定義より $\mathfrak{S} \models \varphi$ と $L(\mathfrak{A}_{\mathfrak{S}}) \subseteq L(\mathfrak{A}_{\varphi})$ は同値である。 $L(\mathfrak{A}_{\varphi}) \cap L(\mathfrak{A}_{\neg \varphi}) = \emptyset$ なので、 $L(\mathfrak{A}_{\mathfrak{S}}) \subseteq L(\mathfrak{A}_{\varphi})$ ならば $L(\mathfrak{A}_{\mathfrak{S}}) \cap L(\mathfrak{A}_{\neg \varphi}) = \emptyset$ である。逆に、 $\mathfrak{S} \not\models \varphi$ 、つまり、 $\mathfrak{S} \models \neg \varphi$ とすると、前述の $\mathfrak{S} \models \varphi$ と $L(\mathfrak{A}_{\mathfrak{S}}) \subseteq L(\mathfrak{A}_{\varphi})$ の同値性より、 $L(\mathfrak{A}_{\mathfrak{S}}) \subseteq L(\mathfrak{A}_{\neg \varphi})$ なので、当然 $L(\mathfrak{A}_{\mathfrak{S}}) \cap L(\mathfrak{A}_{\neg \varphi}) \neq \emptyset$ である。

よって、 $\mathfrak{S} \models \varphi$ と $L(\mathfrak{A}_{\mathfrak{S}}) \cap L(\mathfrak{A}_{\neg \varphi}) = \emptyset$ は同値である。定理 4.7 より $\mathfrak{S} \models \varphi$ は決定可能である。

第5章

効率的な探索

モデル検査をおこなっていくと探索する状態数の増加に苦しめられることが多い。本章では探索する状態数 を削減する方法を紹介する。

5.1 抽象解釈

モデル検査器 SPIN は状態の有限性を前提とした検査をおこなうので、安直に考えると無限の状態を持ってしまうプログラムを検査することはできない。例えば、

```
1 int x;
2
3 active proctype increment()
4 {
5    do
6    :: atomic { x++ }
7    od
8 }
9
10 active proctype decrement()
11 {
12    do
13    :: atomic { x > 0 -> x-- }
14    od
15 }
16
17 ltl { [](0 <= x) }</pre>
```

といったプログラム、つまり、状態遷移系が

$$\cdots \qquad \underbrace{ \begin{cases} s_{-1} \\ \bullet \\ \{x=-1\} \end{cases}}_{\left\{x=0\right\}} \xrightarrow{\begin{cases} s_{0} \\ \bullet \\ \{x=1\} \end{cases}} \xrightarrow{\begin{cases} s_{1} \\ \bullet \\ \{x=2\} \end{cases}} \xrightarrow{\begin{cases} s_{2} \\ \bullet \\ \{x=3\} \end{cases}} \xrightarrow{\left\{x=3\right\}} \cdots$$

$$\mathfrak{S}_{C} = (S_{C}, \rightarrow_{C}, V_{C})$$

$$S_{C} = \{ s_{i} \mid i \in \mathbb{Z} \}$$

$$\rightarrow_{C} = \{ \langle s_{i}, s_{i+1} \rangle \mid 0 \leq i \} \cup \{ \langle s_{i+1}, s_{i} \rangle \mid 0 \leq i \}$$

$$V_{C} = \{ s_{i} \mapsto \{ x = i \} \mid i \in \mathbb{Z} \}$$

であるものを検査すると

といった結果が出力される。errors: 0 で一見うまくいっているように見えるが、1 行目の error を見落としてはならない。探索の深さ上限(デフォルトは 10000)に到達して探索が終了しておらず、その終了していない探索中でエラーが見つかっていないだけである。

このように SPIN は無限の状態を持つプログラムの検査をおこなうことができないが、1.1章で紹介した状態遷移系の定義を読み返すと、そもそも状態集合には集合であることしか要求されておらず、どういった集合をとらなければならないかについては何も要求されていないことに気づく。また、付値の値域中の原子命題についても既に与えられているものと仮定されているだけであり、ここで x=0 といったものをとらなければならないとは言っていない。

そこで、厳密にxの値を見るのでなくxの正負・零非零だけを考慮する状態遷移系を考えてみる。つまり、

$$\mathfrak{S}_{A} = (S_{A}, \rightarrow_{A}, V_{A})$$

$$S_{A} = \{n, z, p\}$$

$$\rightarrow_{A} = \{\langle z, p \rangle, \langle p, z \rangle, \langle p, p \rangle\}$$

$$V_{A} = \{z \mapsto \{x = 0\}, p \mapsto \{x > 0\}, n \mapsto \{x < 0\}\}$$

を考えてみる。この状態線形系の状態数はたったの 3 つである。そのかわりに遷移はややこしくなっており、具体的には自己ループ、つまり、自分自身への遷移が存在する。また、x>0 というように x の値が隠蔽されているため、実際には例えば x=2 であれば遷移は z に遷移するはずはないのであるが z に遷移する可能性があるかもしれないというものになっている。このように G_A は G_C のある種の抽象化になっている。

 G_A は状態数がたった 3 つしかなく SPIN で検査できるので、 G_A を検査することで G_C に対してある種の検査をしたことにできれば話がうまい。そこで、 G_C から G_A へどういった抽象化がおこなわれているかを見ていく。 $\operatorname{sgn}: \mathbb{Z} \to \{\operatorname{n}, \operatorname{z}, \operatorname{p}\}$ を符号判定をする関数とした上で

$$\mathfrak{P}(\mathbb{Z}) \xrightarrow{\alpha} \mathfrak{P}(\{n, z, p\})$$

$$\alpha(C) = \{ \operatorname{sgn}(n) \mid n \in C \}$$
$$\gamma(A) = \{ n \in \mathbb{Z} \mid \operatorname{sgn}(n) \in A \}$$

と定義すると、 α と γ は単調、つまり、任意の $C,C'\subseteq\mathbb{Z}$ に対して $C\subseteq C'$ ならば $\alpha(C)\subseteq\alpha(C')$ であるし、任意の $A,A'\subseteq\{\mathrm{n,z,p}\}$ に対して $A\subseteq A'$ ならば $\alpha(A)\subseteq\alpha(A')$ である。また、 $C\subseteq(\gamma\circ\alpha)(C)$ であり、 $(\alpha\circ\gamma)(A)=A$ である。

この事実について、さらに一般化して、半順序集合 (P,\leq_P) , (Q,\leq_Q) に対して単調な $\alpha\colon (P,\leq_P)\to (Q,\leq_Q)$ と $\gamma\colon (Q,\leq_Q)\to (P,\leq_P)$ が存在し $p\leq_P (\gamma\circ\alpha)(p)$, $(\alpha\circ\gamma)(q)\leq_Q q$ であるとする。 このとき以下が成り立つ。

命題 5.1. $\alpha(p) \leq_Q q$ と $p \leq_P \gamma(q)$ は同値である。

状態遷移系の話に戻る。 $\beta\colon S_C\to S_A$ を

$$\beta(s_i) = \begin{cases} \mathbf{n} & \text{if } i < 0 \\ \mathbf{z} & \text{if } i = 0 \\ \mathbf{p} & \text{if } i > 0 \end{cases}$$

で定義すると、以下のように、この β は遷移を保存する。

命題 5.2. $s_C \to_C s_C'$ ならば $\beta(s_C) \to_A \beta(s_C')$ である。

また、 β を $\beta(X) = \{\beta(x) \mid x \in X\}$ というように集合上に拡張すると以下が成り立つ。

命題 5.3. 任意の $X \subseteq S_C$ に対して $\beta(X) = \alpha(X)$ である。

集合 X を状態集合 S の部分集合とすると X から 1 ステップ以下で到達可能な状態 f(X) は

$$f(X) = X \cup \{ s' \in S \mid$$
ある $s \in X$ があって $s \to s' \}$

である。f は単調である。X から i ステップ以下で到達可能な状態は $f^i(X) = (f \circ \cdots \circ f)(X)$ である。 X から到達可能な状態は $\bigcup \{f^i(X) \mid 1 \leq i\}$ である。

補題 5.4. 任意の $X \subseteq S_C$ に対して $\beta(f_C(X)) \subseteq f_A(\beta(X))$ である。

証明. 命題 5.2 より

$$eta(f_C(X)) = eta(X) \cup \{ eta(s'_C) \in S_A \mid$$
ある $s_C \in X$ があって $s_C \to_C s'_C \}$

$$\subseteq eta(X) \cup \{ s'_A \in S_A \mid$$
ある $s_A \in eta(X)$ があって $s_A \to_A s'_A \}$

$$= f_A(eta(X))$$

である。

定理 5.5. $\bigcup \{f_C^i(X) \mid 1 \leq i\} \subseteq \gamma(\bigcup \{f_A^i(\beta(X)) \mid 1 \leq i\})$ である。

証明. 補題 5.4 に命題 5.3 を通して命題 5.1 を適用することで $f(X) \subseteq \gamma(f_A(\beta(X)))$ を得る。ここで X に $f_C(X)$ を入れることで、 f_A と γ の単調性より、任意の i について $f_C^i(X) \subseteq \gamma(f_A^i(\beta(X)))$ であることがわ かる。 γ は単調なので任意の $i \ge 1$ に対して $\gamma(\{f_A^i(\beta(X)) \mid 1 \le i\}) \subseteq \gamma(\bigcup \{f_A^i(\beta(X)) \mid 1 \le i\})$ である。 ゆえに $\bigcup \{f_C^i(X) \mid 1 \le i\} \subseteq \gamma(\bigcup \{f_A^i(\beta(X)) \mid 1 \le i\})$ である。

定理 5.5 は、 \mathfrak{S}_C における到達可能性(安全性)を知りたいならば \mathfrak{S}_A における到達可能性を調べてその結果を γ に通せばよい、ということを示している。

5.2 半順序簡約

対称性を利用した探索する遷移の削減方法を紹介する。そのために、今までよりも状態遷移系の遷移をもっと細かく見られるようにしておきたい。

本節では、状態遷移系とは $\mathfrak{S}=(S,s_0,T,V)$ という四つ組であり、

- 1. S は有限集合である
- 2. s_0 ∈ S である
- $3.\ T$ は有限集合であり、各 $a\in T$ は S から S への部分関数である
- 4. $V: S \to \mathfrak{P}(AP)$ である

を満たすものをいうことにする。

遷移 $a \in T$ が $s \in S$ で実行可能であるとは $s \in \text{dom}(a)$ であることをいう。実行列 $s_0a_0s_1a_1 \cdots s_{n-1}a_{n-1}s_n$ とは任意の $i \geq 0$ で $a_i(s_i) = s_{i+1}$ をいう。 $a_0a_1 \cdots a_{n-1}$ をこの実行列の遷移列という。 $w = a_0 \cdots a_{n-1}$ と 書くことにしたとき $s_0a_0s_1a_1 \cdots a_{n-1}s_n$ を $s_0 \stackrel{w}{\Rightarrow} s_n$ と書く。

対称的で非反射的な関係 $ID\subseteq T\times T$ が独立関係であるとは、 $s\in S$ で実行可能である $\langle a,b\rangle\in ID$ に対して a が b(s) で実行可能であり、かつ、 $(a\circ b)(s)=(b\circ a)(s)$ であることをいう。

遷移列 $w = a_0 a_1 \cdots$ に対して $\langle a, b \rangle \in ID$ で $ab \equiv_{ID} ba$ を含む最小の同値関係を考える。w の同値類を [w] と書く (Mazurkiewicz 列という)。

例 5.6. $T = \{a, b, c\}$ 、 $ID = \{\langle b, c \rangle, \langle c, b \rangle\}$ とすると w = abca について $[w] = \{abca, acba\}$ である。



命題 5.7. $s_0 \stackrel{w_0}{\Rightarrow} s_n$ かつ $s_0 \stackrel{w_1}{\Rightarrow} s_n'$ かつ $[w_0] = [w_1]$ であるとき $s_n = s_n'$ である。

証明. ID の定義、特に、 $(a \circ b)(s) = (b \circ a)(s)$ から従う。

 s_0 で実行可能な遷移の集合 T が s_0 -永続的であるとは 任意の $0 \le i < n$ で $a_i \not\in T$ である実行列 $s_0a_0s_1a_1\cdots a_{n-1}s_{n-1}a_n$ について、任意の $b\in T$ に対して $\langle a_{n-1},b\rangle\in ID$ であることをいう。 s_0 で実行可能な遷移集合全体 T は、 $a_0\not\in T$ である a_0 が存在しないため、自明に s_0 -永続的である。

 s_0 を初期状態とする状態遷移系 $\mathfrak S$ に対して遷移関係を s_0 -永続的な T_R に制限したものを $\mathfrak S_R$ と書くことにする。このとき以下が成り立つ。

補題 5.8. 状態 s_n を $\mathfrak S$ において任意の $a\in T$ に対して $s_n\not\in \mathrm{dom}(a)$ であるとする。このとき $\mathfrak S$ の実行列 $s_0a_0s_1a_1\cdots s_{n-1}a_{n-1}s_n$ に対してある i が存在して $a_i\in T_R$ である。

証明. いずれも属していないとすると s_0 -永続的の定義より任意の $b\in T_R$ に 対して $\langle a_{n-1},b\rangle\in ID$ でなければならない。独立関係の定義よりある s' が存在して $s_{n-1}a_{n-1}s_nbs'$ という実行列が存在しなければならないが、これは前提に反する。

定理 5.9. 状態 s_n が $\mathfrak S$ において任意の $a\in T$ に対して $s_n\not\in \mathrm{dom}(a)$ であり、 $\mathfrak S$ の実行列 $s_0a_0s_1a_1\cdots s_{n-1}a_{n-1}s_n$ が存在するとき、 s_n は $\mathfrak S_R$ においても s_0 から到達可能である。

証明. 実行列の長さに関する帰納法で示す。実行列の長さが 0 のときは明らかなので実行列の長さ n を n>0 とする。補題 5.8 よりある i が存在して $a_i\in T_R$ である。そういった i のうち最小のものを j とする。任意の $0\le k< j$ について $a_k\not\in T_R$ であることと $a_j\in T_R$ より a_k と a_j は独立なので、 $s_0a_js_1'a_0\cdots s_j$ というように a_j を先頭にもっていった実行列を G においてつくれる。よって、長さ n-1 の G の実行列 $s_1'a_0\cdots s_ja_js_{j+1}a_{j+1}\cdots s_{n-1}a_{n-1}s_n$ に対して存在するので s_1' から s_n へは G_R において到達可能である。よって、 s_0 から s_n に G_R においても到達可能である。

到達可能性についてだけであれば s_0 -永続的な遷移に限っても問題ないことを示した。実際は、示したい性質によっては独立関係をもっと細かくしたものに修正する必要がある。また、 s_0 -永続的な遷移の構成についての説明も省略した。最後に、今回は永続的(persisitency)だけをとりあげたが、ample、stubborn、sleepといったそれぞれ性質の異なるものも知られている。

あとがき

本書はあくまで本トピックの概観を眺めるためのものである。詳細をきちんとおさえるには結局のところ [1] や [2] を読むことになる。もう少し日本語の本を読んでからにしたいというのであれば、良書はたくさんあり、例えば、[3] を勧める。

参考文献

- [1] Edmund M. Clarke, Orna Grumberg, Daniel Kroening, Doron Peled, and Helmut Veith. <u>Model Checking</u>. MIT Press, 2nd edition, 2018.
- [2] Gerard J. Holzmann. The SPIN Model Checker. Addison-Wesley, 2003.
- [3] 萩谷昌己, 吉岡信和, 青木利晃. SPIN による設計モデル検証. 日本評論社, 2008.